



# Modelos dinámicos y computacionales en Economía

## Coordinación en Sistemas Complejos

Licenciatura en Economía, FCEA, UDELAR

7 de noviembre de 2023

## Contenido de la clase:

- Coordinación en los Sistemas Complejos
- Ejemplo: Modelo de Schelling

## Thomas Schelling

- California, 1921 - Maryland, 2016
- Premio Nobel de Economía, 2005 (en conjunto con Robert Aumann)
- 1945 - 1968:
  - Coordinación tácita<sup>1</sup>.
  - Carrera armamentística<sup>2</sup>.
- 1968 en adelante:
  - tipping points<sup>3</sup>.
  - “dying seminar”: influencia de la masa crítica.

---

<sup>1</sup>Schelling, T. C. (1960). *The strategy of conflict*. Harvard University Press.

<sup>2</sup>Schelling, T. C., & Halperin, M. H. (1961). *Strategy and arms control*. Potomac Books Incorporated.

<sup>3</sup>Schelling, T. C. (1978). *Micromotives and macrobehavior*. WW Norton & Company, New York.

## Modelo de Segregación de Schelling (1969<sup>4</sup>, 1971<sup>5</sup>)

- ¿Qué pasaría si todos queremos vivir en un mundo donde al menos un porcentaje de nuestros vecinos comparten alguna característica con nosotros? (aversión a estar en extrema minoría).
- Valor límite: al menos un determinado porcentaje de nuestros vecinos deben ser como nosotros.
- Si este porcentaje es menor que el límite, el individuo se mueve a un lugar vacío (elige el lugar de forma aleatoria).
- Las iteraciones culminan cuando todos cumplen con la restricción.

---

<sup>4</sup>Schelling, T. C. (1969). Models of segregation. The American Economic Review, 59(2), 488-493.

<sup>5</sup>Schelling, T. C. (1971). Dynamic models of segregation. Journal of Mathematical Sociology, 1(2), 143-186.

# Modelo de Segregación de Schelling

## Introducción

- Thomas Schelling hizo los cálculos con peñiques (dorados) y centavos (plateados).
  - En una dimensión: cálculos en el entorno de un punto.
  - En dos dimensiones: tablero de ajedrez.
- En un comienzo<sup>6</sup>, los cálculos se realizaron con lápiz y papel.
- Más adelante en RAND, las simulaciones se realizaron en BASIC.

---

<sup>6</sup>Schelling, T. C. (2006). Some fun, thirty-five years ago. *Handbook of Computational Economics*, 2, 1639-1644.

# Modelo de Segregación de Schelling


¿En qué consiste el modelo?

- 1 Se distribuyen aleatoriamente  $n$  agentes en una grilla bidimensional.
- 2 Cada agente observa si se encuentra “feliz” con su posición en el tablero. Si todos verifican esta condición, el algoritmo se detiene.
- 3 Los que no se encuentran felices, se mueven *aleatoriamente* a otro lugar en el tablero.
- 4 Se repite la recursión (3)-(2) hasta que se detiene el algoritmo.
  - ¿Puede ocurrir que el algoritmo no se detenga? ¿Es necesaria una *stopping rule*?
  - ¿Qué esperamos encontrar cuando la simulación termina?


# Modelo de Segregación de Schelling

<http://ncase.me/polygons-es/>

“Me quiero ir si menos de  $1/3$  de mis vecinos son como yo.”



**insatisfecho:** sólo 1 de 6 de mis vecinos son como yo. menos que  $1/3$ .



**feliz:** 2 de 6 de mis vecinos son como yo. exactamente  $1/3$ .




**indiferente:** todos los vecinos son como yo. (o no tengo vecinos)

# Modelo de Segregación de Schelling

Ingresar a :

**Biblioteca de Modelos / Sample Models / Social Science / Segregation**

A screenshot of the NetLogo Segregation model's code editor. The window title is "Segregation - NetLogo". The menu bar includes "Archivo", "Editar", "Herramientas", "Tamaño", "Pestañas", and "Ayuda". The toolbar has icons for "Buscar...", "Comprobar", and a dropdown menu currently showing "Procedimientos". A checkbox for "Sagrado automático" is checked. The code is organized into sections: "globals" with variables "percent-similar" and "percent-unhappy"; "turtles-own" with variables "happy?", "similar-nearby", "other-nearby", and "total-nearby"; and a "to setup" procedure that includes "clear-all", "ask patches" with a "sprout" loop for creating turtles, and "update-turtles", "update-globals", and "reset-ticks" at the end. The "to setup" section is highlighted in yellow.


```
globals [
  percent-similar ; on the average, what percent of a turtle's neighbors
                  ; are the same color as that turtle?
  percent-unhappy ; what percent of the turtles are unhappy?
]

turtles-own [
  happy?          ; for each turtle, indicates whether at least X-similar-wanted percent of
                  ; that turtle's neighbors are the same color as the turtle
  similar-nearby  ; how many neighboring patches have a turtle with my color?
  other-nearby    ; how many have a turtle of another color?
  total-nearby    ; sum of previous two variables
]

to setup
  clear-all
  ; create turtles on random patches.
  ask patches [
    set pcolor white
    if random 100 < density [ ; set the occupancy density
      sprout 1 [
        ; 105 is the color number for "blue"
        ; 27 is the color number for "orange"
        set color one-of [105 27]
        set size 1
      ]
    ]
  ]
  update-turtles
  update-globals
  reset-ticks
end
```



# Modelo de Segregación de Schelling



```
to go
  if all? turtles [ happy? ] [ stop ]
  move-unhappy-turtles
  update-turtles
  update-globals
  tick
end

to move-unhappy-turtles
  ask turtles with [ not happy? ]
  [ find-new-spot ]
end

to find-new-spot
  rt random-float 360
  fd random-float 10
  if any? other turtles-here [ find-new-spot ] ; keep going until we find an unoccupied patch
  move-to patch-here ; move to center of patch
end

to update-turtles
  ask turtles [
    set similar-nearby count (turtles-on neighbors) with [ color = [ color ] of myself ]
    set other-nearby count (turtles-on neighbors) with [ color != [ color ] of myself ]
    set total-nearby similar-nearby + other-nearby
    set happy? similar-nearby >= ((N-similar-wanted * total-nearby) / 100)
    if visualization = "old" [ set shape "default" set size 1.3 ]
    if visualization = "square-x" [
      ifelse happy? [ set shape "square" ] [ set shape "x" ]
    ]
  ]
end

to update-globals
  let similar-neighbors sum [ similar-nearby ] of turtles
  let total-neighbors sum [ total-nearby ] of turtles
  set percent-similar (similar-neighbors / total-neighbors) * 100
  set percent-unhappy (count turtles with [ not happy? ]) / (count turtles) * 100
end
```

# Modelo de Segregación de Schelling

Variables globales y agentes. Clases y sub-clases.

- Debemos enunciar las variables que utilizaremos en el resto del código.
- No todas las variables afectan a los distintos agentes por igual. Algunas son intrínsecas a cada agente y otras afectan a todos los agentes.
- En este código, las variables globales son promedios en el total de la población o parámetros que se aplican a toda la población.
  - density: Proporción del total de lugares ocupados.
  - %-similar-wanted: Umbral mínimo de vecinos del mismo tipo admisible.
  - percent-similar: Globalmente, ¿cuál es el porcentaje de los vecinos de cada agente que son del mismo color que cada agente?
  - percent-unhappy: ¿qué porcentaje de la población no está conforme con el lugar donde se encuentra?

# Modelo de Segregación de Schelling

Variables globales y agentes. Clases y sub-clases. (cont.)


- Las variables referidas a cada agente “turtles-own” son las siguientes:
  - happy?: si el agente está conforme con su lugar en la grilla.
  - similar-nearby: ¿cuántos en mi entorno son de mi mismo *tipo*?
  - other-nearby: ¿cuántos en mi entorno son *diferentes*?
  - total-nearby: similar-nearby + other-nearby. Total de vecinos.
- Esta es la información que tiene disponible cada agente en este modelo. En este caso, los agentes no interactúan con las variables globales.

# Modelo de Segregación de Schelling

## Procedimientos

- SETUP
- GO
  - move-unhappy-turtles: para indicar que si no están conformes, que utilicen el procedimiento *find-new-spot*.
  - find-new-spot: para buscar un lugar aleatoriamente.
  - update-turtles: calcula si está conforme, a partir de observar su entorno.
  - update-globals: se calculan las variables globales.

# Modelo de Segregación de Schelling



# Modelo de Segregación de Schelling

## Interfaz

- Las variables globales “density” y “%-similar-wanted” pueden modificarse desde la interfaz.
- Los colores simbolizan los *tipos*. Los agentes con forma de “X” son aquellos que no están conformes con el lugar donde se encuentran (en el próximo período, van a modificar su lugar).
- Se muestran gráficos de *percent-similar* y *percent-unhappy*.

# Modelo de Segregación de Schelling


## Resultados:

- ¿qué sucede con un límite alto? (suponemos una densidad del 90% y un valor límite del 65%). ¿qué sucede si llegado el final de la simulación, el umbral pasa al 30%?
- ¿y con límite más bajo? (una densidad del 97% y un valor límite del 30%).
- ¿qué podemos decir acerca de la linealidad de los parámetros?

En el último caso, no hay ningún individuo que busque un vecindario segregado; sin embargo, el resultado global es diferente. Schelling describe este proceso como los "micromotivos" que producen un "macro-comportamiento".

# Modelo de Segregación de Schelling (cont.)


## Ejemplo 1:





# Modelo de Segregación de Schelling (cont.)

## Ejemplo 2:



# modelo de Schelling

## algunos experimentos

- 1 La proporción de individuos insatisfechos disminuye en el tiempo. ¿qué podemos decir acerca de la velocidad? (calcular con un umbral mínimo del 40% y una densidad del 98%) **(ver “experiment1” en BehaviorSpace)**.
- 2 ¿Para qué valores de los parámetros este modelo no converge (establecer un máximo  $t=1000$ )? **(ver “experiment2” en BehaviorSpace)**.
- 3 ¿Los resultados cambian (conformación de grupos bien definidos, segregación) si establecemos un límite máximo a la cantidad de vecinos similares? Establecer valores máximos entre 60% y 90% y mínimos entre 0% y 30%, con una densidad del 85%. **(ver “experiment3” en BehaviorSpace)**.

→ ¿resultados?


# Modelo de Schelling

Algunas ideas:

- Pequeñas tendencias individuales llevan a grandes tendencias colectivas. El problema de la segregación en esta sociedad simulada no es por culpa de todos o de algunos “extremistas”. El “culpable” es el proceso de interacción.
- Alta dependencia temporal. La segregación no disminuye sólo con una caída del umbral, sino que son necesarias otras acciones.
- Demanda por diversidad. Para favorecer a la diversidad, tal vez sea una buena alternativa establecer valores máximos para los *similares* en el entorno.

# Ejemplo: Modelo de Schelling (cont.)

Estudios empíricos surgidos a partir del modelo de Schelling



## Ejemplo: Modelo de Schelling (cont.)

Estudios empíricos surgidos a partir del modelo de Schelling

- Mapa de la segregación en ciudades de Estados Unidos (en base al censo de 2015): <https://www.nytimes.com/interactive/2015/07/08/us/census-race-map.html>
- Mapa de la segregación en ciudades de Estados Unidos (en base al censo de 2020): <https://edition.cnn.com/interactive/2021/us/census-race-ethnicity-map/>
- Para Europa:  
<https://www.bloomberg.com/news/articles/2015-11-16/europe-s-cities-are-suffering-from-economic-segregation-and>
- <https://www.residentialsegregation.org/research-findings>


## Ejemplo: un mercado inmobiliario<sup>7</sup>

- Las mudanzas implican una transacción económica
- Los agentes toman en cuenta el costo de trasladarse a un barrio más afín.
- Se define la utilidad del agente  $i$  con capital  $K_i$  y una propiedad de valor  $P_i$  como  $U_i \sim K_i^\alpha P_i^{1-\alpha}$ .
- El valor que le asigna a una propiedad se encuentra definida por su entorno:
  - $P(X) = A[C(X) - C(\hat{X})] + B$  (alternativa 1)
  - $P(X) = A[C(\hat{X}) - C(X)] + B$  (alternativa 2), donde  $C(X)$  es la cantidad de agentes del color "X" en su vecindad. A y B ctes.
- La dinámica implica que:
  - se eligen dos agentes al azar y cada uno calcula lo que pagaría por la propiedad del otro,
  - se determina un precio de transacción por el promedio
  - intercambian si pueden pagar la diferencia y si ambos mejoran su utilidad.

---

<sup>7</sup>Heymann et al. (2013; pp 85-87). Economía de fronteras abiertas: exploraciones en sistemas sociales complejos. Teseo.

## Ejemplo: un mercado inmobiliario (cont.)



**Figura 2.9:** Evolución del AC de Schelling con la dinámica del mercado inmobiliario. Curvas de nivel de igual utilidad. El AC evolucionó hasta alcanzar un estado asintótico. Los tres experimentos partieron de las mismas condiciones iniciales: además se fijó  $K = 1$  y  $\alpha$  iguales para todos los agentes. Se tomó  $\alpha = 0.1$  para el panel A  $\alpha = 0.5$  para el panel B y  $\alpha = 0.9$  en el Panel C.